

الفصل السادس

دوال البحث والمراجع المتقدمة

**ADVANCED LOOKUP FUNCTIONS**

## 6. دوال البحث المتقدمة

### دالة VLOOKUP

دالة VLOOKUP من أهم دوال البحث الموجودة في الاكسيل حيث تتميز بالسهولة والدقة. وهي تبحث عن قيمة معينة في نطاق البحث بدلالة قيمة موجودة في العمود الأول من نطاق البحث. فلو كان لديك نطاق من البيانات خاص بالموظفين بحيث يظهر فيه رقم الموظف واسمه والقسم الذي يعمل الي آخره من البيانات، فبالإمكان البحث مثلاً عن اسم الموظف أو القسم الذي يعمل به بدلالة رقمه الوظيفي بشرط أن يكون الرقم الوظيفي في العمود الأول من نطاق البحث. والشكل العام لدالة VLOOKUP هو كالتالي:

VLOOKUP(lookup\_value; table\_array; col\_index\_num; [range\_lookup])

حيث أنها تأخذ الوسائط/المعاملات التالية:

- Lookup\_value القيمة المراد البحث بدلالاتها. القيمة التي سيتم البحث عنها في العمود الأول من الجدول أو النطاق.
- Table\_array نطاق البحث
- col\_index\_num رقم العمود المحتوي على نتيجة البحث
- range\_lookup تحدد هل التطابق تام أو تقريبي. إذا كنا نبحث عن تطابق تام فيجب ادخال هذا المعامل كـ 0 أو FALSE وإذا كنا نبحث عن تطابق تقريبي ندخل القيمة 1 أو TRUE. في حال لم يتم ادخال هذا المعامل فان القيمة الافتراضية هي 1.

## ملاحظات/

- إذا كانت قيمة range\_lookup تساوي TRUE أو إذا كانت محذوفة، فيجب وضع قيم العمود الأول لـ table\_array بترتيب فرز تصاعدي؛ وإلا فقد لا تُرجع VLOOKUP القيمة الصحيحة.
- إذا كانت قيمة range\_lookup تساوي FALSE ، فلا داعي لفرز القيم في العمود الأول من table\_array.
- إذا كانت قيمة الوسيطة range\_lookup تساوي FALSE ، ستعثر الدالة VLOOKUP على تطابق تام فقط. وفي حالة تطابق قيمتين أو أكثر في العمود الأول من table\_array مع lookup\_value، فسيتم استخدام أول قيمة يتم العثور عليها. إذا لم يتم العثور على تطابق تام، يتم إرجاع قيمة الخطأ #N/A
- في حال استخدام التطابق التقريبي فان الدالة ترجع أكبر قيمة أقل من أو تساوي القيمة المراد البحث بدلالاتها.

## مثال على استخدام دالة VLOOKUP مع التطابق التام

نطاق البيانات الظاهر يبين لنا رقم الموظف ورقمه والمطلوب هو إيجاد اسم الموظف في الخلية B2 بدلالة رقمه الذي سيتم إدخاله في الخلية B1. للوصول لهذا الهدف نستخدم المعادلة التالية:

=VLOOKUP(B1;D1:E11;2;FALSE)

حيث أن الخلية B1 تحتوي على لقيمة المراد البحث بدلالاتها (رقم الموظف – الذي تم إدخاله في العمود الأول من نطاق البيانات/البحث) و D1:E11 يمثل نطاق البحث، والمعامل الثالث 2 هو رقم العمود المحتوي على نتيجة البحث (العمود الخاص باسم الموظف) والوسيط الأخير قيمته FALSE أو 0 حيث أننا نبحث عن تطابق تام.

B2		: X ✓ fx		=VLOOKUP(B1;D1:E11;2;FALSE)	
	A	B	C	D	E
1	رقم الموظف	1101		رقم الموظف	اسم الموظف
2	اسم الموظف	مهند النعمان		873	حسن م. سالم
3				1109	فريد جمال الدين
4				1549	جمال الأفتدي
5				1334	عماد البيك
6				1643	مصطفى محمد الأغا
7				1101	مهند النعمان
8				1873	ياسر الفرخان
9				983	سعيد محمد العيسى
10				972	سالي الخطيب
11				1398	جورج عيسى

الشكل 6-1

### مثال على استخدام VLOOKUP مع التطابق التقريبي

نطاق البيانات الظاهر يوضح بيانات موظفي إحدى الشركات والمطلوب هو إيجاد نسبة الضريبة لكل موظف. في هذه الحالة لا يمكننا استخدام التطابق التام حيث لا يوجد مبالغ محددة للرواتب وبالتالي نلجأ للتطابق التقريبي.

أولاً تم اعداد نطاق البحث بالشكل الملائم حيث وضعنا قيم الراتب (القيمة المراد البحث بدالاتها) في العمود الأول من نطاق البحث وتم فرز هذا العمود تصاعدياً. في هذه الحالة فإن العمود الأول من نطاق البحث يمثل الشرائح الضريبية فمثلاً الموظف ذو الراتب أقل من 5000 سيطبق عليه نسبة الضريبة 1% و من 5000 حتى أقل من 2500 سيطبق عليه نسبة الضريبة 5% وهكذا.

المعادلة المطلوبة هي كالتالي:

=VLOOKUP(J2;\$L\$1:\$M\$10;2;1)

وفيها يتم البحث في العمود الثاني في النطاق L1:M10 عن أكبر قيمة أقل من أو تساوي القيمة الموجودة في الخلية J2 (الراتب)، تم تثبيت خلايا النطاق \$L\$1:\$M\$10 في المعادلة حتى يتسنى لنا تعميمها على باقي الخلايا.

	A	F	H	I	J	K	L	M
1	Employee Name	Years	Salary	Job Rating	Tax Rate		0	0
2	Page, Lisa	16	60,005	1	8%		5,000	1%
3	Taylor, Hector	4	29,475	4			25,000	5%
4	Dawson, Jonathan	8	46,794	5			35,000	6%
5	Duran, Brian	2	39,248	2			45,000	7%
6	Weber, Larry	7	80,113	2			55,000	8%
7	Pratt, Erik	9	66,913	2			65,000	10%
8	Oconnor, Kent	16	16,764	1			75,000	11%
9	Spencer, Boyd	7	73,238	5			85,000	12%
10	Wiggins, Frank	16	82,665	1			95,000	13%
11	Tanner, Timothy	12	33,858	4				
12	Strickland, Rajean	11	19,509	3				
13	Chase, Troy	10	54,285	4				
14	Brewer, Kent	3	33,490	1				
15	Wilkins, Jesse	2	87,736	5				
	Daniel	14						

الشكل 2-6

### استخدام دالة VLOOKUP مع أكثر من نطاق للبحث

من الممكن استخدام دالة VLOOKUP مع أكثر من نطاق بحث لجعل الدالة أكثر مرونة وفعالية

المطلوب في هذا المثال إيجاد نسبة عمولة البيع لكل مندوب من مندوبي المبيعات اعتماداً على عدد سنوات خدمته حيث أنه هناك جدولين للبحث أحدهما خاص بمندوبي المبيعات ذوي عدد سنوات خدمة أقل من 3 سنين وقد تم تسميته Table1 والآخر خاص بذوي سنوات خدمة أكبر من أو يساوي 3 سنوات وقد تسميته Table2. المعادلة المطلوبة هي كالتالي:

=VLOOKUP(C2;IF(B2<3;Table1;Table2);2)

في هذه المعادلة تم استبدال الوسيط الثاني للدالة (نطاق البحث) بمعادلة IF يتم على أساسها تحديد نطاق البحث المطلوب. يتم ادخال هذه المعادلة في الخلية D2 ومن ثم تعميمها على باقي خلايا حقل نسبة العمولة.

		A		B		C		D		E		F		G		H		I		J		K		L	
		ممثل المبيعات		السنوات		المبيعات		نسبة العمولة		قيمة العمولة				أقل من 3 سنوات		النسبة		أكثر من 3 سنوات		النسبة					
1		Benson	2	120,000	7.00%	8,400								0	1.50%			0	2.00%						
2		Davidson	1	210,921	7.00%	14,764								5,000	3.25%			50,000	6.25%						
3		Ellison	1	100,000	7.00%	7,000								10,000	3.50%			100,000	7.25%						
4		Gomez	2	87,401	6.00%	5,244								20,000	5.00%			200,000	8.25%						
5		Hernandez	6	310,983	9.25%	28,766								50,000	6.00%			300,000	9.25%						
6		Kelly	3	43,902	2.00%	878								100,000	7.00%			500,000	10.00%						
7		Martin	2	121,021	7.00%	8,471								250,000	8.00%										
8		Oswald	3	908	2.00%	18																			
9		Reginald	1	0	1.50%	0																			
10		Veras	4	359,832	9.25%	33,284																			
11		Wilmington	4	502,983	10.00%	50,298																			
12																									
13																									

الشكل 3-6

## دالة MATCH

دالة MATCH ترجع الموقع النسبي لقيمة ما ضمن نطاق/سلسلة من القيم. على سبيل المثال لدينا في الشكل الظاهر نطاق من البيانات في العمود A يشكل مجموعة من أرقام الهويات والمطلوب معرفة الموقع النسبي للقيمة الموجودة في الخلية C2 ضمن هذا النطاق. نستخدم لهذا الغرض دالة Match التالية:

=MATCH(C2;A2:A31;0)

حيث أن المعامل الأول للدالة هو القيمة المراد البحث عنها/بدلاتها والمعامل الثاني عبارة عن نطاق البحث/البيانات و المعامل الثالث يحدد هل التطابق المطلوب تام أو تقريبي وهو يأخذ القيمة 1 أو 0

أو 1-

	A	B	C	D
1	SS#			
2	447-11-6117		560-67-5254	
3	937-98-9910			
4	725-66-0913			
5	744-62-5200			
6	460-05-5559			
7	560-67-5255			
8	295-84-6686			
9	259-14-6820			
10	606-02-4453			
11	882-16-1640			

الشكل 4-6

## ملاحظة:

إذا كانت قيمة الوسيط الثالث هي 1، فإن الدالة تبحث عن أكبر قيمة تكون أصغر من أو تساوي القيمة التي يتم البحث عنها (المعامل الأول) وفي هذه الحالة يجب ترتيب النطاق بشكل تصاعدي. وإذا كانت القيمة 0 فإن الدالة Match تبحث عن أول قيمة تكون مطابقة تماما للقيمة التي يتم البحث عنها. وإذا كانت القيمة -1 فإن الدالة MATCH تبحث عن أصغر قيمة تكون أكبر من أو تساوي القيمة التي يتم البحث عنها وفي هذه الحالة يجب ترتيب النطاق بشكل تنازلي. إذا تجاهلت الوسيط الثالث فإن الدالة تفترضه 1.

### استخدام دالة Match لتحديد هل القيمة المحددة موجودة ضمن نطاق محدد أم لا

هذا المثال هو تعديل للمثال السابق و المطلوب هو تحديد هل الرقم الموجود في الخلية C2 موجود ضمن الأرقام الموجودة في العمود A

الفكرة هي أن الدالة MATCH اذا لم تجد تطابق فإنها ترجع قيمة الخطأ #NA وبالتالي سوف نستخدم دالة IF التي سوف تفحص هل ناتج الدالة MATCH هل هو خطأ أم لا. في حال كانت النتيجة خطأ فان الرقم "غير موجود" والا فانه موجود وعلى ذلك نكتب المعادلة التالية:

=IF(ISERROR(MATCH(C2;A:A;0));"موجود";"غير موجود")

حيث استخدمنا دالة ISERROR لفحص هل نتيجة الدالة MATCH خطأ أم لا.

### دالة INDEX

دالة INDEX ترجع القيمة الموجودة في موقع نسبي معين في نطاق محدد. فمثلا إذا كنا نريد القيمة التي تقع في الموقع الرابع من النطاق C3:C7 (شكل 5-6) نكتب الدالة التالية:

=INDEX(C3:C7;4)

حيث أن الدالة INDEX تأخذ ثلاث معاملات. المعامل الأول عبارة عن نطاق البيانات التي تقع القيمة المراد ارجاعها بداخله، و المعامل الثاني هو رقم الصف للقيمة المرادة و المعامل الثالث وهو معامل اختياري عبارة عن رقم العمود للقيمة المراد ارجاعها وبما أن نطاق البيانات في مثالنا عبارة عن عمود واحد فقط فلا يلزم استخدام المعامل الثالث. أما لو أردنا الحصول على القيمة الموجودة في لصف الثالث والعمود الرابع من النطاق C3:J7 فنكتب المعادلة التالية:

=INDEX(C3:J7;3;4)

حيث أننا استخدمنا المعامل الثالث في هذه الحالة.

	A	B	C	D	E	F	G	H	I	J
1			منطقة الشحن							
2			1	2	3	4	5	6	7	8
3		1	10.69	12.79	15.29	18.29	21.89	26.29	31.49	37.79
4		2	12.29	14.69	17.59	21.09	25.29	30.29	36.29	43.49
5		3	14.19	16.99	20.39	24.49	29.39	35.29	42.29	50.69
6		4	16.89	20.29	24.29	29.09	34.89	41.89	50.29	60.29
7		5	18.29	21.89	26.29	31.49	37.79	45.29	54.29	65.09

الشكل 5-6

مثال: استخدام دالة INDEX لاستخراج قيمة معينة من نطاق ما

في ورقة العمل الظاهرة (شكل 6-6) نطاق البيانات C3:J7 يوضح تكلفة الشحن بناء على منطقة الشحن وحجم العنصر المراد شحنه. والمطلوب هو الحصول على تكلفة شحن العناصر الموجودة في العمود P حيث انه في العمودين Q و R يظهر حجم العنصر و منطقة الشحن الخاصة به. للوصول للهدف نكتب المعادلة التالية:

$$=INDEX(\$C\$3:\$J\$7;Q2;R2)$$

حيث المعامل الأول هو نطاق البحث المحتوي على أسعار الشحن و المعامل الثاني عبارة عن رقم الصف المحتوي على النتيجة والثالث رقم العمود. ثم نعمم الصيغة على باقي الخلايا.

		A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S
1		منطقة الشحن														الحجم رقم الصنف	منطقة الشحن	تكلفة الشحن		
2			1	2	3	4	5	6	7	8							406B92	5	4	31.49
3		1	10.69	12.79	15.29	18.29	21.89	26.29	31.49	37.79							429A85	1	3	
4		2	12.29	14.69	17.59	21.09	25.29	30.29	36.29	43.49							124B35	4	2	
5		3	14.19	16.99	20.39	24.49	29.39	35.29	42.29	50.69							964C47	4	2	
6		4	16.89	20.29	24.29	29.09	34.89	41.89	50.29	60.29							760H13	2	2	
7		5	18.29	21.89	26.29	31.49	37.79	45.29	54.29	65.09							143H70	4	5	
8																	331R75	5	3	

الشكل 6-6

### استخدام دالتي INDEX-MATCH لإنشاء عمليات بحث فعالة ومرنة

تعرفنا على كيفية استخدام دالتي INDEX و MATCH بشكل منفصل ورأينا استخداما مفيدا لكل منهما ولكن قوة هاتين الدالتين الحقيقية تكمن في استخدامهما مع بعضهما البعض.

#### مثال على إجراء البحث باستخدام INDEX-MATCH

في الشكل 6-7 لدينا نطاق من البيانات يتكون من حقلين الحقل الأول خاص برقم الضمان الاجتماعي والثاني خاص بالاسم والمطلوب هو اظهار اسم الشخص ذو رقم الضمان الموجود في الخلية P2 وذلك في الخلية Q2 .

وللوصول لهذه النتيجة يجب أن نحدد أولا الموقع النسبي لرقم الضمان في عمود ارقام الضمان وهو العمود M وذلك من خلال دالة MATCH التالية:

=MATCH(P2;M:M;0)

ثم نستخدم دالة INDEX للحصول على الاسم الموجود في الموقع الذي حصلنا عليه من خلال دالة MATCH وذلك من قائمة الأسماء الموجودة في العمود N. من خلال المعادلة التالية:

=INDEX(N:N;MATCH(P2;M:M;0))

=INDEX(N:N;MATCH(P2:M:M;O))				
M	N	O	P	Q
SS#	الاسم		SS#	الاسم
447-11-6117	Booth, Ron		560-67-5255	Collins, Quinn
937-98-9910	Brennan, Tito			
725-66-0913	Caballero, Richard			
744-62-5200	Carmichael, George			
460-05-5559	Christinelli, David			
560-67-5255	Collins, Quinn			
295-84-6686	Hanks, Sally			
259-14-6820	Hendricks, Anne			
606-02-4453	Hewitt, Steven			

الشكل 7-6

### ملاحظة:

يمكن الحصول على نفس النتيجة السابقة من خلال دالة VLOOKUP ولكن استخدام دالتي INDEX-MATCH أكثر مرونة حيث ان دالة VLOOKUP تشترط أن القيمة المراد البحث بدالاتها يجب أن تكون في العمود الأول من نطاق البحث بينما هذا القيد غير موجود في INDEX-MATCH.

### تنفيذ البحث ثنائي الاتجاه من خلال دالتي INDEX-MATCH

في هذا المثال لدينا نطاق من البيانات يوضح لنا مبيعات مجموعة من المنتجات في مجموعة من الأشهر والمطلوب هو إيجاد مبيعات منتج محدد لشهر محدد. ونحن قد سمحنا للمستخدم بإدخال المنتج والشهر المطلوبين في الخلايا B1 و B2 والمطلوب إيجاد المبيعات في الخلية B8.

	A	B	D	E	F	G	H	I	J	K
1	الشهر	مارس			Disk Drives	CD Drives	DVD Drives	Flash Drives	Combined	
2	المنتج	CD Drives		يناير	5,760	3,740	5,029	4,718	19,247	
3				فبراير	6,739	2,739	4,993	2,615	17,086	
4	انزياح الشهر			مارس	6,338	3,358	4,265	5,312	19,273	
5	انزياح المنتج			أبريل	5,013	4,866	4,565	1,108	15,552	
6	المبيعات			مايو	6,204	2,048	4,766	1,994	15,012	
7				يونيو	6,522	2,842	5,379	3,830	18,573	
8	باستخدام معادلة واحدة			يوليه	6,456	2,714	4,171	3,232	16,573	
9				أغسطس	6,836	2,081	5,077	1,607	15,601	
10				سبتمبر	5,967	4,574	4,754	1,563	16,858	
11				أكتوبر	6,576	4,008	4,830	2,590	18,004	
12				نوفمبر	6,042	4,235	4,224	3,960	18,461	
13				ديسمبر	5,566	3,751	5,103	3,013	17,433	
14				الإجمالي	74,019	40,956	57,156	35,542	207,673	
15										

الشكل 8-6

وللوصول الى هذه النتيجة يجب ان نحدد أولاً رقم الصف الذي يقع فيه الشهر المطلوب (انزياح الشهر) ورقم لعمود الذي يقع به المنتج (انزياح المنتج).

للحصول على انزياح الشهر نستخدم دالة MATCH التالية في الخلية B4:

=MATCH(B1;E2:E14;0)

حيث أن B1 تمثل الشهر المطلوب و E2:E12 يمثل نطاق الأشهر.

للحصول على انزياح المنتج نستخدم دالة MATCH في الخلية B5:

=MATCH(B2;F1:J1;0)

حيث أن B2 تحتوي على المنتج المطلوب و F1:J1 يمثل نطاق المنتجات.

الآن بعد الحصول على العمود و الصف اللذان تقع النتيجة عند تقاطعهما في الخلايا B4 و B5 نستخدم دالة INDEX في الخلية B6 للحصول على مبيعات المنتج الموجود في الخلية B1 للشهر في B2 كالتالي:

=INDEX(F2:J14;B4;B5)

حيث F2:J14 هو نطاق البحث .

وللحصول على النتيجة من خلال معادلة واحدة نكتب المعادلة التالية في الخلية B8:

=INDEX(F2:J14;MATCH(B1;E2:E14;0);MATCH(B2;F1:J1;0))

	A	B	C	D	E	F	G	H	I	J
1	الشهر	مارس				Disk Drives	CD Drives	DVD Drives	Flash Drives	Combined
2	المنتج	CD Drives			يناير	5,760	3,740	5,029	4,718	19,247
3					فبراير	6,739	2,739	4,993	2,615	17,086
4	انزياح الشهر	3			مارس	6,338	3,358	4,265	5,312	19,273
5	انزياح المنتج	2			أبريل	5,013	4,866	4,565	1,108	15,552
6	المبيعات	3,358			مايو	6,204	2,048	4,766	1,994	15,012
7					يونيو	6,522	2,842	5,379	3,830	18,573
8	باستخدام معادلة واحدة	3,358			يوليه	6,456	2,714	4,171	3,232	16,573
9					أغسطس	6,836	2,081	5,077	1,607	15,601
10					سبتمبر	5,967	4,574	4,754	1,563	16,858
11					أكتوبر	6,576	4,008	4,830	2,590	18,004
12					نوفمبر	6,042	4,235	4,224	3,960	18,461
13					ديسمبر	5,566	3,751	5,103	3,013	17,433
14					الإجمالي	74,019	40,956	57,156	35,542	207,673

الشكل 6-9

### إجراء البحث ثنائي الأعمدة باستخدام دالتي INDEX-MATCH

أحيانا قد تكون مضطراً لإجراء عملية البحث بناء على قيم موجودة في أكثر من حقل في نطاق البحث في نفس اللحظة. فعلى سبيل المثال في ورقة العمل الظاهرة (شكل 6-11) نطاق البيانات D1:F12 يوضح لنا مجموعة من أنواع السيارات و موديلاتها و الكود الخاص بكل سيارة و المطلوب هو الحصول على كود السيارة التي نوعها و موديلها مدخلين في الخليتين B1:B2 ففي هذه الحالة المطلوب إجراء عملية البحث بناء على قيمتين في حقلين مختلفين و ليس قيمة واحدة كما هي العادة.

	A	B	C	D	E	F	G
1	النوع	Toyota		النوع	الموديل	النوع و الموديل	الكود
2	الموديل	Sequoia		Chevy	Suburban		C-094
3	الكود			Chevy	Tahoe		C-823
4				Ford	Explorer		F-772
5				Ford	Escape		F-229
6				Honda	Pilot		I-897
7				Honda	CR-V		I-900
8				Jeep	Compass		J-983
9				Jeep	Grand Cherokee		J-701
10				Nissan	Suburban		N-231
11				Toyota	Sequoia		T-871
12				Toyota	Land Cruiser		T-981
13							

الشكل 11-6

الشكل 6-10

يمكن اجراء هذه العملية بأكثر من طريقة

**الطريقة الأولى** هي ادراج حقل جديد في نطاق البيانات محتوياته هي عبارة عن

محتويات حقل النوع والموديل كما هو ظاهر في الشكل 6-12 ومن ثم سوف نستخدم هذا الحقل للبحث على أساسه. حسب المعادلة التالية:

=INDEX(G1:G12;MATCH(B1&B2;F1:F12;0))

الشكل 12-6

	A	B	C	D	E	F	G
1	النوع	Toyota		النوع	الموديل	النوع و الموديل	الكود
2	الموديل	Sequoia		Chevy	Suburban	=D2&E2	C-094
3	الكود			Chevy	Tahoe		C-823
4				Ford	Explorer		F-772
5				Ford	Escape		F-229
6				Honda	Pilot		I-897
7				Honda	CR-V		I-900
8				Jeep	Compass		J-983
9				Jeep	Grand Cherokee		J-701
10				Nissan	Suburban		N-231
11				Toyota	Sequoia		T-871
12				Toyota	Land Cruiser		T-981

في هذه المعادلة استخدمنا دالة MATCH لتحديد موقع الكود المطلوب النسبي ضمن قائمة الأكواد ومن ثم دالة INDEX لاستخراج هذا الكود بناء على موقعه الذي حصلنا عليه من MATCH.

لنتوقف قليلاً عند دالة MATCH الداخلية MATCH(B1&B2;F1:F12;0) في هذه الدالة المعامل الأول (القيمة المراد البحث بدالاتها) عبارة عن النوع والموديل معا (قيمتين من حقلين مختلفين) تم ربطهما معا باستخدام معامل الارتباط &. ونطاق البحث F1:F12 هو عبارة عن حاصل ربط حقلي النوع والموديل كما رأينا سابقاً.

	A	B	C	D	E	F	G
1	النوع	Toyota		النوع	الموديل	النوع و الموديل	الكود
2	الموديل	Sequoia		Chevy	Suburban	ChevySuburban	C-094
3	الكود			Chevy	Tahoe	ChevyTahoe	C-823
4				Ford	Explorer	FordExplorer	F-772
5				Ford	Escape	FordEscape	F-229
6				Honda	Pilot	HondaPilot	I-897
7				Honda	CR-V	HondaCR-V	I-900
8				Jeep	Compass	JeepCompass	J-983
9				Jeep	Grand Cherokee	JeepGrand Cherokee	J-701
10				Nissan	Suburban	NissanSuburban	N-231
11				Toyota	Sequoia	ToyotaSequoia	T-871
12				Toyota	Land Cruiser	ToyotaLand Cruiser	T-981

الشكل 6-13

**الطريقة الثانية** لاجراء البحث ثنائي الأعمدة هي باستخدام صيغ المصفوفات وفيها سوف نستغني عن الحقل المرحلي الذي أضفناه في الطريقة الأولى (النوع والموديل) عن طريق انشاء هذا الحقل في الذاكرة. وذلك حسب معادلة لمصفوفات التالية:

=INDEX(F1:F12;MATCH(B1&B2;D1:D12&E1:E12;0))

ولا تنسى أن تضغط Ctrl+Shift+Enter لإدخال هذه المعادلة.

هذه المعادلة هي نفس المعادلة السابقة ولكن نطاق البحث بالنسبة لدالة MATCH حصلنا عليه من خلال المعادلة D1:D12&E1:E12 والتي سوف ينتج عنها مصفوفة في الذاكرة سوف تستخدمها دالة MATCH كنطاق للبحث.

		B3 : {=INDEX(F1:F12;MATCH(B1&B2;D1:D12&E1:E12;0))}				
	A	B	C	D	E	F
1	النوع	Toyota		النوع	الموديل	الكود
2	الموديل	Sequoia		Chevy	Suburban	C-094
3	الكود	T-871		Chevy	Tahoe	C-823
4				Ford	Explorer	F-772
5				Ford	Escape	F-229
6				Honda	Pilot	I-897
7				Honda	CR-V	I-900
8				Jeep	Compass	J-983
9				Jeep	Grand Cherokee	J-701
10				Nissan	Suburban	N-231
11				Toyota	Sequoia	T-871
12				Toyota	Land Cruiser	T-981
13						

الشكل 14-6

### استخدام دالة INDEX مع أكثر من نطاق للبيانات

دالة INDEX لها حالتين (كما بالشكل 15-6 الحالة الأولى تتعامل مع نطاق بيانات واحد وهي الحالة التي تعاملنا معها في الأمثلة السابقة والحالة الثانية تتعامل مع أكثر من نطاق في نفس الوقت وتأخذ البنينة و المعاملات/الوسائط التالية:

**INDEX(reference, row\_num, [column\_num], [area\_num])**

الشكل 6-15

=INDEX(				
INDEX(array; row_num; [column_num])				
INDEX(reference; row_num; [column_num]; [area_num])				

يحتوي بناء جملة الدالة INDEX على الوسيطات التالية:

- Reference وسيطة مطلوبة. مرجع إلى نطاق خلايا واحد أو أكثر.
- Row\_num وسيطة مطلوبة. رقم الصف في المرجع الذي سيتم إرجاع مرجع منه.
- Column\_num وسيطة اختيارية. رقم العمود في المرجع الذي سيتم إرجاع مرجع منه.
- Area\_num وسيطة اختيارية. تحديد نطاق في مرجع سيتم إرجاع تقاطع Row\_num و Column\_num منه. يتم ترقيم الناحية الأولى التي حددها أو أدخلتها بالرقم 1، والثانية بالرقم 2، وهكذا. إذا تم حذف Area\_num، فتستخدم الدالة INDEX الناحية 1.

على سبيل المثال، إذا وصفت الوسيطة Reference الخلايا (A1:B4,D1:E4,G1:H4)، فتكون Area\_num 1 النطاق A1:B4، وتكون Area\_num 2 النطاق D1:E4 وتكون Area\_num 3 النطاق G1:H4

#### مثال على استخدام INDEX مع أكثر من نطاق

ورقة العمل الظاهرة في الشكل 6-16 توضح لنا مبيعات مجموعة من المحاصيل لأشهر معينة لعدة مناطق والمطلوب إيجاد مبيعات محصول محدد لشهر محدد ومنطقة محددة. فهذا يكون المطلوب بحث ثنائي الاتجاه متعدد النطاقات. نستخدم لهذا الغرض دالة INDEX متعددة النطاقات كالتالي (مع العلم بأننا أسمينا نطاقات المحاصيل كالتالي: banana, pear, and apple):

```
=INDEX((banana;pear;apple);MATCH(N5;A1:A6;0);MATCH(N4;A1:J1;0);IF(N3="banana";1;IF(N3="pear";2;3)))
```

حيث أننا أدخلنا أسماء نطاقات البحث كوسيط/معامل أول للدالة INDEX، الوسيط الثاني و الثالث (رقم الصف و العمود) حصلنا عليهم من خلال دالة MATCH كما هو ظاهر في المعادلة، أما الوسيط الأخير للدالة و الذي من خلاله نحدد للدالة نطاق البحث المطلوب فحددناه من خلال معادلة IF والتي سوف ترجع لنا القيمة 1 اذا كان المحصول Banana و القيمة 2 لمحصول Pear و القيمة 3 لمحصول Apple.

فمثلا اذا كان المحصول المدخل في الخلية N3 هو Apple فان دالة IF سوف ترجع الرقم 3 و بالتالي فنطاق البحث المطلوب البحث بداخله هو النطاق الثالث Apple.

	A	B	C	D	E	F	G	H	I	J	K	L	N	O
1	Banana	Jan	Feb	Mar	QTR 1	Apr	May	Jun	QTR 2	H1				
2	North	34	43	65	142	34	45	23	102	244				
3	South	23	23	32	78	24	25	21	70	148			apple	مبيعات
4	East	3	2	5	10	5	3	7	15	25			may	لشهر
5	West	12	12	17	41	15	15	13	43	84			west	و منطقة
6	Total	72	80	119	271	78	88	64	230	501				
7														
8	Pear	Jan	Feb	Mar	QTR 1	Apr	May	Jun	QTR 2	H1			54	هي
9	North	23	21	34	78	35	29	43	107	185				
10	South	34	35	43	112	27	28	26	81	193				
11	East	19	21	25	65	21	25	24	70	135				
12	West	11	15	18	44	18	21	23	62	106				
13	Total	87	92	120	299	101	103	116	320	619				
14														
15	Apple	Jan	Feb	Mar	QTR 1	Apr	May	Jun	QTR 2	H1				
16	North	12	14	15	41	18	17	15	50	91				
17	South	2	5	8	15	4	6	122	132	147				
18	East	3	5	2	10	5	6	7	18	28				
19	West	44	55	63	162	33	54	61	148	310				
20	Total	61	79	88	228	60	83	205	348	576				

الشكل 6-16

## دالة INDIRECT

دالة INDIRECT من الدوال القوية والمرنة التي نستطيع من خلالها انشاء معادلات قوية ومرنة . وهي تأخذ نص كوسيط وترجع المرجع الذي له نفس الاسم. فمثلا الدالة التالية:

=INDIRECT("A2")

تقيم النص المدخل لها (الوسيط) فترجع المرجع الذي له نفس الاسم وهو A2 وبالتالي سوف تظهر محتويات الخلية A2 في الخلية التي كتبنا بها دالة INDIRECT.

فلو طبقنا الدالة السابقة في الخلية J2 في ورقة العمل الظاهرة في الشكل فسترجع لنا ما في الخلية A2.

	A	B	C	D	E	F	G	H	I	J
1	ABC Corporation - Sales by Product Type									
2	Summary	Tables	Chairs	Desks	Couches	Beds	Total	% of Total		Summary
3	Retail	4,850	5,410	4,320	4,770	5,660	25,010	37.4%		
4	Wholesale	2,710	3,250	3,520	4,390	3,500	17,370	25.9%		
5	Internet	4,080	4,880	4,680	5,350	5,580	24,570	36.7%		

الشكل 17-6

ومن الممكن استخدام معامل الربط & للحصول على النص Reference text الذي نريد استخدامه كوسيط للدالة INDIRECT كمثل على ذلك:

=INDIRECT(J4&J3)

فتكون النتيجة هو محتوى الخلية B3

	A	B	C	D	E	F	G	H	I	J	K
1	ABC Corporation - Sales by Product Type										
2	Summary	Tables	Chairs	Desks	Couches	Beds	Total	% of Total			
3	Retail	4,850	5,410	4,320	4,770	5,660	25,010	37.4%		3	
4	Wholesale	2,710	3,250	3,520	4,390	3,500	17,370	25.9%		b	
5	Internet	4,080	4,880	4,680	5,350	5,580	24,570	36.7%		=INDIRECT(J4&J3)	
6	Total	11,640	13,540	12,520	14,510	14,740	66,950	100.0%		INDIRECT(ref_text; [a1])	

الشكل 6-18

## مثال : عرض بيانات تلخيصيه من عدة أوراق عمل بشكل مرن

في المصنف الظاهر في الشكل 6-19 لدينا مجموعة من أوراق العمل بحيث تحتوي كل ورقة عمل المبيعات الخاصة بمنطقة معينة وورقة العمل المسماة INDIRECT-1 تحتوي على مجموع المبيعات للمناطق الأربعة.

والمطلوب هو أن نعرض مجموع مبيعات كل منطقة في صفحة الملخص INDIRECT-1 بحيث نكتب المعادلة الأولى في الخلية B9 ثم نعمم على الخلايا B10:B12.

الخلية التي تحتوي على مجموع مبيعات منطقة معينة هي الخلية G6 من ورقة العمل الخاصة بهذه المنطقة فمثلا مجموع مبيعات المنطقة الشرقية East من الممكن الحصول عليه من خلال المعادلة التالية:

=East!G6

ABC Corporation - Sales by Product Type							
Summary	Tables	Chairs	Desks	Couches	Beds	Total	% of Total
Retail	4,850	5,410	4,320	4,770	5,660	25,010	37.4%
Wholesale	2,710	3,250	3,520	4,390	3,500	17,370	25.9%
Internet	4,080	4,880	4,680	5,350	5,580	24,570	36.7%
<b>Total</b>	<b>11,640</b>	<b>13,540</b>	<b>12,520</b>	<b>14,510</b>	<b>14,740</b>	<b>66,950</b>	<b>100.0%</b>
% of Total	17.4%	20.2%	18.7%	21.7%	22.0%	100.0%	
East							
South							
Midwest							
West							

ABC Corporation - Sales by Product Type							
East	Tables	Chairs	Desks	Couches	Beds	Total	% of Total
Retail	1,130	1,210	1,320	960	1,150	5,770	32.1%
Wholesale	850	920	1,030	1,330	980	5,110	28.4%
Internet	1,090	1,410	1,420	1,640	1,540	7,100	39.5%
<b>Total</b>	<b>3,070</b>	<b>3,540</b>	<b>3,770</b>	<b>3,930</b>	<b>3,670</b>	<b>17,980</b>	<b>100.0%</b>
% of Total	17.1%	19.7%	21.0%	21.9%	20.4%	100.0%	

الشكل 6-19

وبالتالي بالإمكان كتابة المعادلة التالية للحصول على مجموع مبيعات للمنطقة الشرقية:

`=INDIRECT(A9&"!"&"g6")`

ومن ثم تعميمها على باقي الخلايا للحصول على النتيجة المطلوبة. لاحظ هنا انه لو كان لدينا 50 أو 100 منطقة فسيكون المطلوب هو فقط التعميم على باقي الخلايا وهذا بسبب المرونة العالية لدالة .INDIRECT

SUM : X ✓ fx =INDIRECT(A9&"!"&"g6")								
	A	B	C	D	E	F	G	H
1	ABC Corporation - Sales by Product Type							
2	Summary	Tables	Chairs	Desks	Couches	Beds	Total	% of Total
3	Retail	4,850	5,410	4,320	4,770	5,660	25,010	37.4%
4	Wholesale	2,710	3,250	3,520	4,390	3,500	17,370	25.9%
5	Internet	4,080	4,880	4,680	5,350	5,580	24,570	36.7%
6	Total	11,640	13,540	12,520	14,510	14,740	66,950	100.0%
7	% of Total	17.4%	20.2%	18.7%	21.7%	22.0%	100.0%	
8								
9	East	=INDIRECT(A9&"!"&"g6")			=East!G6			
10	South	INDIRECT(ref_text; [a1])						
11	Midwest							
12	West							

الشكل 20-6


## مثال 2: استخدام دالة INDIRECT لإنشاء قائمة منسدلة معتمدة على قائمة أخرى

في ورقة العمل الظاهرة لدينا نطاق ممن البيانات يوضح مجموعة من أسماء الولايات وأسماء المدن التابعة لكل ولاية والمطلوب هو انشاء قائمة بأسماء الولايات في الخلية C2 ثم انشاء قائمة منسدلة في الخلية D2 تظهر أسماء المدن الخاصة بالولاية الظاهرة في الخلية C2.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2			الولاية			الولاية	أسماء المدن لكل ولاية						
3		WI	Green Bay		AZ	Chandler	Gilbert	Glendale	Mesa	Peoria	Phoenix	Scottsdale	
4		Green Bay			CA	Anaheim	Bakersfield	Fresno	Long Beach	Los Angeles	Oakland	Sacramento	
5		Madison			CO	Arvada	Aurora	Boulder	Colorado Springs	Denver	Fort Collins	Lakewood	
6		Milwaukee			IA	Cedar Rapids	Des Moines						
7					IL	Aurora	Chicago	Joliet	Naperville	Peoria	Rockford	Springfield	
8					IN	Evansville	Fort Wayne	Indianapolis	South Bend				
9					KS	Kansas City	Olathe	Overland Park	Topeka	Wichita			
10					MO	Independence	Kansas City	Springfield	St Louis				
11					NE	Lincoln	Omaha						
12					OH	Akron	Cincinnati	Cleveland	Columbus	Dayton	Toledo		
13					UT	Provo	Salt Lake City	West Valley City					
14					WI	Green Bay	Madison	Milwaukee					


الشكل 21-6

لهذا الهدف نسمي أولاً نطاقات الخلايا المحتوية على أسماء المدن لكل ولاية بحيث مثلاً نطاق الخلايا الذي يحتوي على أسماء المدن الخاصة بولاية أريزونا G3:Q3 ب AZ. وأسهل طريقة لتسمية الخلايا هي تحديد نطاق الولايات والمدن F3:Q14 ثم الذهاب الى شريط Formulas ثم الى Defined names (الأسماء المعرفة) ثم Create From Selection (الانشاء من التحديد) فيظهر مربع الحوار الخاص بإنشاء أسماء من التحديد نحدد الخيار الخاص ب Left column (العمود الايسر) حتى يأخذ كل صف الاسم الموجود في الخلية الموجودة في أقصى يساره.



الشكل 22-6

ولإنشاء قائمة منسدلة في الخلية B3 تسرد أسماء الولايات نقف في الخلية B3 ثم نذهب الى شريط Data (بيانات) ثم Data Validation (التحقق من البيانات) من قائمة Allow (السماح) نختار List (قائمة) ونحدد عناصر القائمة بالنطاق .F3:F14.




الشكل 23-6

الآن لإنشاء قائمة منسدلة في الخلية C3 معتمدة على القيمة الظاهرة في B3 نقف بداخل الخلية C3 ثم نذهب الى Data validation نختار List من قائمة Allow وفي مربع النص الخاص ب source نكتب المعادلة التالية:

=INDIRECT(B3)

حيث أن دالة INDIRECT هنا ترجع المرجع الذي له نفس الاسم الموجود في الخلية B3. وللتخلص من الفراغات التي قد تظهر في بعض قوائم المدن، نحدد نطاق الولايات والمدن ثم نذهب الى Home ثم Find and select (بحث وتحديد) ثم Go to special (الذهاب الى خاص) ثم نختار Blanks (الفراغات). نضغط موافق ثم نضغط بزر الماوس الأيمن ونختار Delete.



الشكل 6-24

## استخدام دالة INDIRECT لتنفيذ البحث ثنائي الاتجاه

سابقاً نحن أجرينا عملية البحث ثنائي الاتجاه من خلال دالتي INDEX-MATCH وفي هذا المثال سنتعرف على كيفية الحصول على نفس النتيجة بصورة أسهل من خلال استخدام دالة INDIRECT

المثال:

نطاق البيانات الظاهر يوضح مبيعات مجموعة من المنتجات لمجموعة من الأشهر و المطلوب هو ارجاع مبيعات منتج معين لشهر محدد.

	A	B	C	D	E	F	G	H	I
1	Month:	July			Widgets	Sprockets	Snapholytes	Combined	
2	Product:	Sprockets	January	2,892	1,771	4,718	9,381		
3			February	3,380	4,711	2,615	10,706		
4			March	3,744	3,223	5,312	12,279		
5			April	3,221	2,438	1,108	6,767		
6			May	4,839	1,999	1,994	8,832		
7			June	3,767	5,140	3,830	12,737		
8			July	5,467	3,337	3,232	12,036		
9	Single-formula -->	3,337	August	3,154	4,895	1,607	9,656		
10			September	1,718	2,040	1,563	5,321		
11			October	1,548	1,061	2,590	5,199		
12			November	5,083	3,558	3,960	12,601		
13			December	5,753	2,839	3,013	11,605		
14			Total	44,566	37,012	35,542	117,120		
15									
16									

الشكل 25-6

الفكرة هي أنه بإمكاننا الحصول على النتيجة المطلوبة من خلال إيجاد تقاطع الصف المحتوي على الشهر المطلوب مع العمود المحتوي على المنتج المطلوب.

وللقيام بهذه العملية نسمي الصفوف والأعمدة بحيث أن يأخذ كل صف اسم الشهر الخاص به وكل عمود اسم المنتج الخاص به. وأسهل طريقة لتسمية الصفوف والأعمدة هي كالتالي:

- بالنسبة للصفوف نحدد النطاق D2:H14 ثم نذهب الى تبويب Formulas ثم الى الأسماء المعرفة Definbed Names نضغط على انشاء من التحديد Create From Selection ثم نحدد على مربع الاختيار Left Column.

- بالنسبة للأعمدة نحدد النطاق E1:H14 ثم نذهب إلى Formulas ثم Create From Selection ثم نحدد مربع الاختيار Top Row.

ثم نستخدم المعادلة التالية للحصول على النتيجة:

=INDIRECT(B1) INDIRECT(B2)

في هذه المعادلة استخدمنا دالة INDIRECT لتحويل ما يوجد في الخلايا B1 و B2 إلى ما يقابله من نطاق ثم استخدمنا معامل التقاطع وهو المسافة (" ") للحصول على تقاطع صف الشهر المحدد (الموجود في B1) مع عمود المنتج المطلوب (الموجود في B2).

### دالة OFFSET

دالة OFFSET هي أحد دوال المراجع وهي ترجع مرجع لنطاق يتكون من عدد محدد من الصفوف و الأعمدة وبيعد عدد معين من الصفوف و الأعمدة عن خلية محددة أو نطاق من الخلايا. وتأخذ البنية التالية:

OFFSET(reference, rows, cols, [height], [width])

يحتوي بناء جملة الدالة OFFSET على الوسائط/المعاملات التالية:

- Reference مطلوبة. المرجع الذي تريد أن تستند منه الإزاحة. يجب أن يشير المرجع إلى خلية ؛ بخلاف ذلك، تُرجع الدالة OFFSET قيمة الخطأ #VALUE!
- Rows مطلوبة. عدد الصفوف التي يبعد عنها النطاق المطلوب تحديده/ارجاعه عن الخلية المرجع. اذا كان الرقم بالموجب فهذا يعني ان التحرك يتم للأسفل واذا كان بالسالب فالتحرك للأعلى.
- Cols عدد الأعمدة التي يبعد عنها النطاق المطلوب عن الخلية المرجع. اذا كان بالموجب فان التحرك لليمين واذا كان بالسالب فالتحرك لليسار (في ورقة عمل اتجاهها من اليسار لليمين).
- Height اختيارية. الارتفاع، في عدد الصفوف، الذي تريده للنطاق الذي يتم إرجاعه.
- Width اختيارية. العرض، في عدد الأعمدة، الذي تريده للنطاق الذي يتم إرجاعه. يجب

فلو أردنا معرفة محتوى الخلية التي تبعد أربعة صفوف للأسفل من الخلية A1 في الشكل 6-26 بالإمكان استخدام الدالة التالية:

=OFFSET(A1;4;0)

حيث أن الوسيط الأول A1 هي الخلية المرجع، الوسيط الثاني هو عدد الصفوف الذي تبعد الخلية المطلوب ارجاعها بمقداره عن الخلية المرجع وبما انه الاتجاه للأسفل استخدمنا علامة الموجب، الوسيط الثالث هو 0 لأننا نريد التحرك على نفس العمود، الوسيطين الرابع و الخامس بالإمكان الاستغناء عنهما في هذه المثال حيث أن النطاق المطلوب ارجاعه يتكون من خلية واحدة و بالتالي لا يوجد له طول وعرض.

	A	B	C	D	E
	Reporting Date	Closing Rate		Last Reporting Date	Latest Closing Rate
1					
2	06/05/2013	234.20		15/05/2013	256.00
3	07/05/2013	238.00			
4	08/05/2013	244.50			
5	09/05/2013	246.70			
6	10/05/2013	242.00		=OFFSET(A1;4;0)	
7	13/05/2013	244.50			
8	14/05/2013	253.80			
9	15/05/2013	256.00			

الشكل 6-26

ومثال آخر لو أردنا معرفة مجموع النطاق المكون من 3 صفوف وعمود واحد ويبعد عن الخلية B1 بمقدار 4 صفوف على نفس العمود نستخدم المعادلة التالية:

=SUM(OFFSET(B1;4;0;3;1))

حيث أننا حصلنا أولاً على النطاق المطلوب من خلال دالة OFFSET ثم أدخلنا هذا النطاق على دالة SUM للحصول على النتيجة المطلوبة.

ولنأخذ مثلاً آخر على دالة OFFSET حيث المطلوب هو ارجاع التاريخ الأخير في العمود A وذلك في الخلية D2. ونستخدم لذلك المعادلة التالية:

=OFFSET(A1;COUNTA(A:A)-1;0)

	A	B	C	D	E
	Reporting Date	Closing Rate		Last Reporting Date	Latest Closing Rate
1					
2	06/05/2013	234.20		=OFFSET(A1;COUNTA(A:A)-1;0)	
3	07/05/2013	238.00			
4	08/05/2013	244.50			
5	09/05/2013	246.70			
6	10/05/2013	242.00			
7	13/05/2013	244.50			
8	14/05/2013	253.80			
9	15/05/2013	256.00			

الشكل 6-27

حيث استخدمنا A1 كخلية مرجع، عدد الصفوف الذي نريد أن نتحركه هو عدد غير ثابت وبالتالي حصلنا عليه من خلال معادلة هي COUNTA(A:A)-1 والتي يتم فيها حساب عدد الخلايا غير

الفارغة في العمود A مطروحاً منه واحد لأننا نريد العد من بعد الخلية A1. واستخدمنا 0 للوسيط الثالث لأننا نتحرك على نفس العمود، وتم إهمال الوسيطين الرابع والخامس لأن النطاق المُرجع عبارة عن خلية واحدة.

ومثال آخر: لو أردنا أن نعرف متوسط سعر الاغلاق لأخر ثلاث أيام نستخدم المعادلة التالية:

=AVERAGE(OFFSET(B1;COUNTA(B:B)-1;0;-3;1))

حيث أننا استخدمنا دالة OFFSET لإرجاع النطاق المطلوب إيجاد متوسط خلاياه كالتالي:

OFFSET(B1;COUNTA(B:B)-1;0;-3;1)

وفيها تم استخدام B1 كخلية مرجع، وحصلنا على عدد الأعمدة الذي نريد أن نتحركه بدءاً من خلية المرجع من خلال معادلة COUNTA(B:B)-1، الوسيط الثالث 0 لأننا نتحرك على نفس العمود، الوسيط الرابع هو 3 وعلامته سالب لأننا تحركنا حتى الخلية الأخيرة في العمود B (في حالتنا هي الخلية B9) وكي نحدد النطاق المطلوب نريد التحرك للأعلى ثلاث صفوف وأن يكون عدد أعمدة النطاق المُرجع 1 (الوسيط الخامس).

	A	B	C	D	E	F	G	H
	Reporting Date	Closing Rate		Last Reporting Date	Latest Closing Rate	Average of last 3 days	Average of last 5 days	
1								
2	06/05/2013	234.20		15/05/2013		=AVERAGE(OFFSET(B1;COUNTA(B:B)-1;0;-3;1))		
3	07/05/2013	238.00						
4	08/05/2013	244.50						
5	09/05/2013	246.70						
6	10/05/2013	242.00						
7	13/05/2013	244.50						
8	14/05/2013	253.80						
9	15/05/2013	256.00						

الشكل 6-28

### دالة CHOOSE

دالة CHOOSE هي احدى دوال البحث والمراجع وهي ترجع قيمة من بين مجموعة من القيم بدلالة موقع هذه القيمة بالنسبة لباقي القيم. وهي تأخذ البنية التالية:

CHOOSE(index\_num; value1; [value2], ...)

حيث أن الوسيط الأول للدالة index\_num يمثل رقم القيمة المطلوب ارجاعها بالنسبة لمجموعة القيم المعطاة. باقي الوسائط تمثل مجموعة القيم، تستطيع دالة CHOOSE أن تأخذ حتى 254 قيمة.

على سبيل المثال الدالة التالية:

=CHOOSE(3;B6;C6;D6;E6;F6;G6)

ترجع قيمة الخلية التي ترتيبها 3 من بين القيم الموجودة وهي 119.

	A	B	C	D	E	F	G	H
1	Banana	Jan	Feb	Mar	Apr	May	Jun	
2	North	34	43	65	34	45	23	
3	South	23	23	32	24	25	21	
4	East	3	2	5	5	3	7	
5	West	12	12	17	15	15	13	
6	Total	72	80	119	78	88	64	
7								
8								
9								
10					=CHOOSE(3;B6;C6;D6;E6;F6;G6)			
11					CHOOSE(index_num; value1; [value2]; [value3]			

الشكل 6-29

### مثال على استخدام دالة CHOOSE

نطاق البيانات الظاهر في شكل 6-30 يوضح لنا مجموع مبيعات محصول معين لمجموعة من الأشهر، والمطلوب هو إيجاد مجموع المبيعات حتى شهر محدد (سوف يتم إدخاله في الخلية K4) وذلك في الخلية K8.

وللحصول على هذه النتيجة نستخدم المعادلة التالية:

$$=SUM(B6:CHOOSE(K4;B6;C6;D6;E6;F6;G6))$$

حيث أن نطاق الجمع في هذه المعادلة يمتد من الخلية B6 والتي تحتوي على مجموع المبيعات لشهر يناير وحتى الخلية التي تحتوي على مجموع مبيعات الشهر المحدد في الخلية K4، هذه الخلية حصلنا عليها من خلال دالة CHOOSE التالية (CHOOSE(K4;B6;C6;D6;E6;F6;G6) حيث أن الوسيط الأول لهذه الدالة هو رقم القيمة المطلوب إرجاعها (وهي القيمة في الخلية K4) وباقي وسائط الدالة هي عبارة عن الخلايا المحتوية على مبيعات كل شهر من الأشهر.

	A	B	C	D	E	F	G	H	K	L
1	Banana	Jan	Feb	Mar	Apr	May	Jun			
2	North	34	43	65	34	45	23			
3	South	23	23	32	24	25	21		banana	مبيعات
4	East	3	2	5	5	3	7		2	لشهر
5	West	12	12	17	15	15	13			
6	Total	72	80	119	78	88	64			
7										
8									152	هي
9										

الشكل 6-30

ومن الممكن جعل ورقة العمل أكثر فاعلية عن طريق تحويل الخلية K2 إلى قائمة منسدلة تحتوي على أسماء الأشهر ومن ثم تعديل معادلة الناتج لتصبح كالتالي:

$$=SUM(B6:CHOOSE(MATCH(K4;N3:N14;0);B6;C6;D6;E6;F6;G6))$$

	A	B	C	D	E	F	G	H	K	L	M	N
1	Banana	Jan	Feb	Mar	Apr	May	Jun					
2	North	34	43	65	34	45	23					
3	South	23	23	32	24	25	21		banana	مبيعات		jan
4	East	3	2	5	5	3	7		apr	لشهر		feb
5	West	12	12	17	15	15	13		apr			mar
6	Total	72	80	119	78	88	64		may			apr
7									jun			may
8									jul			jun
9									aug			jul
10									sep			aug
11									oct	349	هي	sep
12									nov			oct
13												nov
14												dec
15												

الشكل 6-31

وهي نفس المعادلة السابقة ولكننا بدلاً من استخدام رقم الشهر بشكل صريح استخدمنا اسم الشهر ومن ثم استخدمنا دالة MATCH للحصول على رقم الشهر.

## ملاحظات:

- للمزيد حول دالة MATCH يرجى مراجعة الفقرة الخاصة بهذه الدالة في هذا الفصل.
- لإنشاء قائمة منسدلة بالأشهر أدخلنا أسماء الأشهر في النطاق N3:N14 ثم استخدمنا ميزة التحقق من البيانات Data Validation للحصول على قائمة منسدلة بناء على القيم في خلايا نطاق الأشهر من خلال الذهاب إلى تبويب Data Validation ثم Data Validation ثم من مربع الحوار الخاص بـ Data Validation نحدد List في خانة Allow وندخل النطاق المحتوي على أسماء الأشهر في خانة Source (شكل 6-32)

الشكل 32-6

